

## 3 Gebundene Bewegungen

### 3.3 Bewegung entlang einer Zykloiden

Im letzten Teil sahen wir, dass die Periode eines an eine Kreisbahn gebundenen Pendels von der Amplitude abhängt. Es gibt jedoch eine spezielle Bahnform, bei der die Periode eines sich hin und her bewegendes Massepunktes nicht von der Amplitude abhängig ist. Diese Bahn oder Kurve ist als Zykloide bekannt.




Fig.3.3-1

Eine Zykloide ist eine geometrische Kurve, die von einem Punkt P am Umfang eines Kreises beschrieben wird, wenn der Kreis entlang einer Geraden rollt, ohne zu gleiten. (Beispielsweise beschreibt das Pedal eines Fahrrades eine Zykloide. Wenn ein Kreis auf einem anderen Kreis rollt, nennt man die Kurve eines Punktes, der fest mit dem rollenden Kreis verbunden ist, Epizykloide oder Hypozykloide.)

Es handelt sich bei der Zykloide um eine nicht algebraische Kurve, die bereits 1599 von Galileo Galilei untersucht und benannt wurde. Die Achse des rollenden Kreises, bzw. Scheibe, bewegt sich mit der Geschwindigkeit  $v = -\omega a$ , wobei  $a$  der Radius der rollenden Scheibe und  $\omega$  ihre Winkelgeschwindigkeit sind. Das Minuszeichen tritt deshalb auf, weil die Winkelgeschwindigkeit im Uhrzeigersinn negativ ist. Der Punkt B an der Oberseite der Scheibe hat die Geschwindigkeit  $v - \omega a = 2v$ . Der B gegenüberliegende Punkt der Scheibe hat die Geschwindigkeit null, da die Scheibe nicht gleiten soll. Der Kontaktpunkt der Scheibe mit der Rollbahn ist in jedem Augenblick in Ruhe. (Man zeigt geometrisch, dass eine Tangente in P auch durch B geht. Genauere Berechnungen in 3.3.3. Auch im Internet finden Sie zahlreiche Artikel zur Zykloide. Besonders schön ist die Seite <http://www.mathcurve.com/index.htm>)

Wir können die Figur 3.3-1 auch mit MuPAD zeichnen. Dazu ist es vorteilhaft, die x-y-Koordinaten des Punktes P mit Hilfe von Radius  $a$  und Rollwinkel  $\theta = \text{PCM}$  darzustellen. Aus der obigen Skizze ersehen wir, dass folgende Beziehungen gelten

$$\begin{aligned}x &= ON = OM - NM = a(\theta - \sin(\theta)) \\y &= NP = MC + AC = a(1 - \cos(\theta))\end{aligned}\tag{3.3-1}$$

Wir zeichnen zunächst einfach eine Zykloide und den dazugehörigen Kreis, der der x-Achse entlang rollt.

```
plot(plot::Curve2d([t-sin(t), 1-cos(t)],
t = 0..2*PI),
plot::Point2d([t-sin(t),1-cos(t)], t = 0..2*PI,
PointSize = 2*unit::mm,Color=RGB::Red),
plot::Circle2d(1, [t,1],t = 0..2*PI),
Scaling=Constrained)
```




Fig.3.3-2

Für die **Animation** machen wir einen Unterschied zwischen dem Radius  $r$  des Rollkreises und der Entfernung  $a$  des Punktes  $P$  vom Mittelpunkt  $C$ , denn der Punkt  $P$  kann sich z.B. im Innenbereich der rollenden Scheibe befinden, etwa bei  $a = 0.8$  und  $r = 1$ , vgl. Fig.3.3-3. Für diesen allgemeineren Fall mit  $a \neq r$  lauten die Gleichungen 3.3.1

$$\begin{aligned}x(\theta) &= r\theta - a \sin(\theta) \\y(\theta) &= r - a \cos(\theta)\end{aligned}$$

In den Programmen schreiben wir  $t$  für  $\theta$ , d.h.  $t$  bedeutet hier nicht die Zeit. (Im Normalfall bedeutet  $t$  die Zeit und es gilt  $\theta = \omega t$ .)

```
reset():
a:=0.8: r:=1:
/*"reduzierte Zykloide", weil a < r;
a ist die Entfernung PC und r ist der radius des Kreises.
Man waehlt r fest und variiert a: a=r oder a>r oder a<r*/
plot(plot::Curve2d([r*t-a*sin(t),r-a*cos(t)], t=0..tmax,
tmax = 0..3*PI),// t=Theta
plot::Point2d([r*t-a*sin(t),r-a*cos(t)],
t = 0..3*PI,
```

```

PointSize = 2*unit::mm,Color=RGB::Green),
plot::Circle2d(r, [t,r],Color=RGB::OliveGreenDark,
t = 0..3*PI),
plot::Line2d([t,r],[r*t-a*sin(t),r-a*cos(t)],
Color=RGB::Red,t = 0..3*PI),
Scaling=Constrained)

```




Fig.3.3-3

Wir sehen, wie die "verkürzte" Zyklode beim Abrollen des Kreises entsteht.

Bei der "invertierten" Zyklode der Fig.3.3-4 sehen wir einen Massepunkt, der reibungsfrei eine Zyklode hinabgleitet und hin und her schwingt. Auch werden die Vektoren von Geschwindigkeit und Beschleunigung eingetragen. Vgl. Programm 3 in Abschnitt 2.2.

```

x:=t->t+sin(t):
//Die invertierte Zyklode hat x = r(theta+sin(theta))
y:=t->1-cos(t)//r=a=1
curve:=plot::Curve2d([x(t),y(t)],t=-PI..PI):
pos:=t->(x(t),y(t))//Vektor Position
vel:=t->(x'(t),y'(t))//Vektor Geschwindigkeit
acel:=t->(x''(t),y''(t))//Beschleunigung
scale:=1//legt die Pfeillaenge fest
x1:=pos(t1)[1]//Anfang des Vektors (x'(t),y'(t))
y1:=pos(t1)[2]:
v1:=vel(t1)[1]//Geschw. des Punktes (x(t),y(t))
v2:=vel(t1)[2]:
a1:=acel(t1)[1]//Beschleunigung
a2:=acel(t1)[2]:
x2:=x1+v1*scale//Spitze des Vektors (x'(t),y'(t))
y2:=y1+v2*scale:
x3:=x1+a1*scale//Spitze des Vektors (x''(t),y''(t))
y3:=y1+a2*scale:
p:=plot::Point2d([x1,y1],t1=-PI..PI,


```

```

Color=RGB::Black,
PointSize=3*unit::mm):
ve:=plot::Arrow2d([x1,y1],[x2,y2],
t1=-PI..PI,Color=RGB::Red):
ac:=plot::Arrow2d([x1,y1],[x3,y3],
t1=-PI..PI,Color=RGB::Green):
plot(curve,p,ac,ve,
Header="Ein Massepunkt gleitet auf einer Zykliden",
HeaderFont=["Arial",12,Bold],
HeaderAlignment=Center,
Footer="Gruen = Beschleunigung,
Rot = Geschwindigkeit",
FooterAlignment=Center,
Scaling = Constrained)

```

### Ein Massepunkt gleitet auf einer Zykliden



Gruen = Beschleunigung,  
Rot = Geschwindigkeit

Fig.3.3-4

#### 3.3.1 Das Zykliden-Pendel

Der Oszillator in Fig. 3.3-4 besteht aus einem Massepunkt  $m$ , der reibungsfrei unter dem Einfluss der Schwerkraft auf einer invertierten Zyklode gleitet. Die Amplitude der Schwingung hängt von dem Punkt ab, in dem  $m$  freigelassen wird. Es ist jedoch interessant, dass die Periode immer gleich  $T = 4\pi(r/g)^{1/2}$  ist, also unabhängig ist von der Amplitude;  $r$  ist der Radius des Kreises, der die Zyklode erzeugt. (Es war Christian Huygens, *Horologium oscillatorium*, 1673, der eine praktische Methode erfand, mit der er ein Zykliden-Pendel herstellen konnte.)




Fig.3.3-5

Auf die Masse  $m$ , die im Punkt A ohne Anfangsgeschwindigkeit freigelassen wurde, wirken zwei Kräfte: die Bindungskraft  $\vec{N}$  und die Schwerkraft  $m\vec{g}$ . Die Kraft  $\vec{N}$  steht immer senkrecht auf der Bindung (die Zykloide),  $m\vec{g}$  zeigt immer nach unten (in die  $-\vec{j}$ -Richtung). Die Vektoren  $\vec{n}$  und  $\vec{t}$  sind Einheitsvektoren in Richtung von  $\vec{N}$  und in Richtung der Tangente. Das zweite Newtonsche Gesetz liefert uns zwei skalare Gleichungen

$$m \frac{dv}{dt} = -mg \sin(\tau) \quad (3.3-2)$$

$$mv^2/\rho = N - mg \cos(\tau) \quad (3.3-3)$$

$dv/dt$  ist der Betrag der tangentialen Beschleunigung,  $\rho$  ist der Krümmungsradius, und  $v^2/\rho$  ist der Betrag der Normalbeschleunigung. Wir kommen in 3.3.3 auf diese Gleichungen zurück.

Die Fig. 3.3-6 illustriert die Huygenssche Idee. Eine Masse, am unteren Ende eines Fadens, wird von zwei zykloidenförmigen "Backen" gezwungen, auf einer Zykloidenbahn zu schwingen, und zwar mit einer Periode, die nicht von der Amplitude abhängt. (Dass die Bahn wirklich eine Zykloide ist, kann man geometrisch zeigen.) Für kleine Schwingungen haben die Backen keinen großen Einfluss auf die Schwingungen, und die Periode ist praktisch gleich der eines mathematischen Pendels:  $T = 2\pi(l/g)^{1/2}$ . Dasselbe Resultat ergibt sich aus der Gleichung  $T = 4\pi(r/g)^{1/2}$ , denn im Falle des Zykloidenpendels haben wir  $l = 4r$ . Schauen Sie sich nun die folgende Illustration an:

```

reset()//Zykloidenpendel
plot(plot::Curve2d([t-sin(t)+PI,-(1-cos(t))],
Color=RGB::Black,t=-2*PI..0),
plot::Point2d([t-sin(t)+PI,-(1-cos(t))], t = -PI-0.2..0,
PointSize = 3*unit::mm,Color=RGB::Red),
plot::Line2d([t-sin(t)+PI,-(1-cos(t))],[0,2],

```

```

Color=RGB::Blue,t = -PI-0.2..0),
plot::Curve2d([t-sin(t),-(1-cos(t)-2)],Color=RGB::Green,
t = -PI..PI),
Scaling=Constrained)

```




Fig.3.3-6

### 3.3.2 Übungen

Rein mechanisch kann man eine *normale* Zykloide ( $a = r$ ) mit Hilfe eines Kugelschreibers aufzeichnen, der im Punkt P an einer festen Scheibe mit Radius  $r$  befestigt ist. Die Scheibe rollt ohne zu gleiten über eine glatte, feste Unterlage. Wenn die Entfernung  $a$  des Punktes P vom Mittelpunkt der Scheibe größer als  $r$  sein sollte, entsteht eine *verschlungene* Zykloide. Falls  $a < r$ , ergibt sich eine *gestreckte* Zykloide. Die parametrischen Gleichungen lauten in diesen Fällen

$$x = rt - a \sin(t) \text{ und } y = r - a \cos(t)$$

Entwirf ein Programm, das diese Fälle darstellt.

**Lösung:**

```

reset();//Das Programm gilt fuer alle drei Faelle
a:=1.4:r:=1.0://a=r ou a<r ou a>r
plot(
plot::Curve2d([r*t-a*sin(t),r-a*cos(t)],t=0..tmax,
tmax = 0..3*PI),
plot::Point2d([r*t-a*sin(t),r-a*cos(t)], t = 0..3*PI,
PointSize = 2*unit::mm,Color=RGB::Red),
plot::Line2d([t,r],[r*t-a*sin(t),r-a*cos(t)],
Color=RGB::Green,t = 0..3*PI),
plot::Circle2d(r, [t,r], t = 0..3*PI),
Scaling=Constrained)

```




Fig.3.3.7

### Die Epizykloide

entsteht, wenn die Scheibe über einen festen Kreis vom Radius  $R$  rollt. Die Koordinaten des Punktes  $P$  sind jetzt

$$x = (R + r) \cos(t) - a \cos[t(R + r)/r] \text{ und } y = (R + r) \sin(t) - a \sin[t(R + r)/r]$$

**Die Kardioid** ist eine Epizykloide mit  $r = R = a$ .

Die Länge dieser Kurve ist  $L = 8r$ , die Fläche beträgt  $A = 6r^2\pi$ , d.h. sechsmal so groß wie die Fläche der Scheibe.

### Die Hypozykloide

Bei ihr rollt die Scheibe im Innern eines festen Kreises. Die Parameter  $a$  und  $r$  ändern ihre Vorzeichen:

$$x = (R - r) \cos(t) + a \cos[(R - r)/r] \text{ und } y = (R - r) \sin(t) - a \sin[t(R - r)/r]$$

Wenn wir in diesen Gleichungen  $4r = R$  setzen, ergibt sich eine **Asteroid** (Sternkurve) mit der Darstellung

$$x = R \cos^3(t) \text{ und } y = R \sin^3(t)$$

Wenn wir den Parameter  $t$  eliminieren, erhalten wir

$$x^{2/3} + y^{2/3} = R^{2/3}$$

Alle Kurven sollen graphisch dargestellt werden.

### Lösungen

Epizykloide

```

reset()//Epizykloide
a:=1.0:r:=1.0:R:=4.0://a=r=1 und R:r = 4
plot(
plot::Curve2d([(R+r)*cos(t)-a*cos(t*(R+r)/r),
(R+r)*sin(t)-a*sin(t*(R+r)/r)],t=0..tmax, tmax = 0..2*PI),
plot::Point2d([(R+r)*cos(t) - a*cos(t*(R+r)/r),
(R+r)*sin(t)-a*sin(t*(R+r)/r)], t = 0..2*PI,
PointSize = 2*unit::mm,Color=RGB::Red),
plot::Line2d([(R+r)*cos(t),
(R+r)*sin(t)],[(R+r)*cos(t)-a*cos(t*(R+r)/r),
(R+r)*sin(t)-a*sin(t*(R+r)/r)],
Color=RGB::Green,t = 0..2*PI),
plot::Circle2d(R, [0,0],Color=RGB::Black, t = 0..2*PI),
plot::Circle2d(r, [(R+r)*cos(t),
(R+r)*sin(t)],Color=RGB::Black, t = 0..2*PI),
Scaling=Constrained)

```

Epizykloide mit  $a=r=1, R:r=4$




Fig.3.3-8

Epizykloide mit  $a=1.5, r=1, R:r=3$






Fig.3.3-9

### Hypozykloide

```

reset()//Hypozykloide
a:=1.5:r:=1.0:R:=4.0://a>r und R:r = 4
plot(
plot::Curve2d([(R-r)*cos(t)+a*cos(t*(R-r)/r),
(R-r)*sin(t)-a*sin(t*(R-r)/r)],t=0..tmax, tmax = 0..2*PI),
plot::Point2d([(R-r)*cos(t) + a*cos(t*(R-r)/r),
(R-r)*sin(t)-a*sin(t*(R-r)/r)], t = 0..2*PI,
PointSize = 2*unit::mm,Color=RGB::Red),
plot::Line2d([(R-r)*cos(t),
(R-r)*sin(t)],[(R-r)*cos(t)+a*cos(t*(R-r)/r),
(R-r)*sin(t)-a*sin(t*(R-r)/r)],
Color=RGB::Green,t = 0..2*PI),
plot::Circle2d(R, [0,0],Color=RGB::Black, t = 0..2*PI),
plot::Circle2d(r, [(R-r)*cos(t),
(R-r)*sin(t)],Color=RGB::Black, t = 0..2*PI),
Scaling=Constrained)

```




Fig.3.3-10

Asteroide (Hypozykloide mit  $4r=R$  und  $a=r=1$ ):




Fig.3.3-11

### 3.3.3 Lösung mit Bleistift und Papier

Wir wollen uns jetzt ein wenig mit der Theorie der vorgestellten Kurven beschäftigen. Die Gleichungen (3.3-1) sind eine Parameterdarstellung einer Zyklode für den Fall  $a = r$ . In den Animationen haben wir die folgende allgemeinere Form benutzt

$$\begin{aligned}x(\theta) &= r\omega\theta - a\sin(\omega\theta) \\y(\theta) &= r - a\cos(\omega\theta)\end{aligned}$$

Die Winkelgeschwindigkeit  $\omega$  haben wir immer gleich 1rad/s gewählt. Aus Fig. 3.3-1 entnehmen wir, dass  $PC = BC$ , woraus folgt, dass Winkel  $PBC =$  Winkel  $BPC$ . Anhand der Winkel im Dreieck  $PBC$  sehen wir, dass  $\tau = (\pi - \theta)/2$ .

Der Punkt  $P$  nimmt gleichzeitig an zwei Geschwindigkeiten teil: eine tangential zum rollenden Kreis, eine zweite horizontal zur  $x$ -Achse. Der Betrag beider Geschwindigkeiten ist  $v = r\omega = r\frac{d\phi}{dt}$ . Der resultierende Geschwindigkeitsvektor geht durch die Punkte  $P$  und  $B$ , d.h. er liegt auf der Tangente an die Zyklode im Punkt  $P$ . Wenn wir in die Tangentengleichung  $y = mx + n$  die Koordinaten der Punkte  $P$  und  $B$  einsetzen, also  $P(r(\theta - \sin(\theta)), r(1 - \cos(\theta)))$  und  $B(r\theta, 2r)$ , erhalten wir  $m = (1 + \cos(\theta))/\sin(\theta) = \cot(\theta/2)$ . Also

$$m := \tan(\tau) = \cot(\theta/2) \tag{3.3-4}$$

Es ist lehrreich, sich diese Gleichung anhand der Definition  $m = \tan(\tau) = \frac{dy}{dx}$  herzuleiten. Im Fall  $a = r$  erhalten wir aus (3.3-1) die Beziehungen  $\frac{dx}{d\theta} = x'(\theta) = r - r\cos(\theta) = r(1 - \cos(\theta))$  und  $\frac{dy}{d\theta} = y'(\theta) = r\sin(\theta)$ . Daraus folgt  $\frac{dy}{dx} = \frac{dy}{d\theta} \frac{d\theta}{dx} = \sin(\theta)/(1 - \cos(\theta)) = \cot(\theta/2)$ , also erneut (3.3-4).

Um den Umfang eines Zykloidenbogens ( $\theta$  von 0 bis  $2\pi$ ) zu berechnen, benötigen wir das Bogenlängenelement  $ds = (dx^2 + dy^2)^{1/2}$ . Wir substituieren  $dx$  und  $dy$  durch  $dx = r(1 - \cos(\theta))d\theta$  und  $dy = r \sin(\theta)d\theta$  und erhalten

$$ds = 2r \sin(\theta/2)d\theta \quad (3.3-5)$$

Wenn wir diesen Ausdruck zwischen 0 und  $2\pi$  integrieren, erhalten wir für die Länge eines vollen Zykloidenbogens (bei  $a = r$ ) den achtfachen Radius des erzeugenden Kreises, denn

$$s = 2r \int_0^{2\pi} \sin(t/2)dt = -4r [\cos(t/2)]_0^{2\pi} = 8r$$

Es ist interessant, dass  $\pi$  nicht im Ergebnis erscheint!

Die Fläche zwischen diesem Bogen und der x-Achse ist dreimal so groß wie die Fläche des rollenden Kreises, denn

$$A = \int_{\phi=0}^{2\pi} ydx = r^2 \int_0^{2\pi} (1 - \cos(t) - \cos^2(t))dt = 3r^2\pi$$

Wir haben dabei folgende Transformation benutzt:  $ydx = y(t)x'(t)dt = r^2(1 - \cos(t))^2$

Mit Hilfe der Integrationsprozedur `int` berechnen wir die Integrale auch sehr leicht mit MUPAD :

```
f:=2*r*sin(t/2):
Laenge:=int(f,t=0..2*PI)
```

$$8 \cdot r$$

```
f:=r^2*(1-cos(t))^2:
Flaech:=int(f,t=0..2*PI)
```

$$3 \cdot \pi \cdot r^2$$

Wenn wir eine ausführlichere Rechnung bevorzugen, schreiben wir

```
x:=r*(t-sin(t)):
y:=r*(1-cos(t)):
Flaech:=int(y*diff(x,t),t=0..2*PI)
```

$$3 \cdot \pi \cdot r^2$$

In der Gleichung (3.3-3) erschien der *Krümmungsradius*  $\rho$ . Man versteht darunter den Betrag der Ableitung  $ds/d\tau$ . In den folgenden Abschnitten werden wir diesen Begriff oft benutzen. Bei der Zykloide erhalten wir für die Ableitung  $ds/d\tau$  den Wert  $4r \cos(\tau)$ , denn  $ds/d\tau = ds/d\theta \cdot d\theta/d\tau = 2r \cos(\theta/2) \cdot 2 = 4r \cos(\tau)$ .

Der *Berührungskreis* an eine Kurve in einem gegebenen Punkt A ist derjenige Kreis, der sich dort am besten an die Kurve anschmiegt. Der reziproke Wert  $\kappa = 1/\rho$  ist die *Krümmung* der Kurve in A. Wenn die Kurve selbst ein Kreis ist, stimmt  $\rho$  mit dem Kreisradius  $r$  überein. In diesem Fall ist die Zentripetalbeschleunigung gegeben durch  $a_n = v^2/\rho = v^2/r$ , vgl. Gleichung (3.3-3).

Zum Abschluss dieses Abschnitts wollen wir die **Periode**  $T$  eines Zykloidenpendels berechnen und zeigen, dass  $T$  unabhängig ist von der Amplitude des Pendels. Dafür brauchen wir zunächst einen geeigneten Ausdruck für die Geschwindigkeit. Wir beginnen mit

$$\frac{dv}{dt} = \frac{dv}{ds} \frac{ds}{dt} = \frac{dv}{ds} v = \frac{d(v^2/2)}{ds} \quad (3.3-7)$$

Aus der Gleichung (3.3-2),  $mdv/dt = -mg \sin(\tau)$ , erhalten wir  $dv/dt = -g \sin(\tau) = -\frac{g}{4r} s$ , wenn wir Gl. (3.3-6) benutzen. Die Integration von  $\frac{d(v^2)}{ds} = -\frac{g}{2r} s$  ergibt

$$v^2 = -\frac{g}{4r} s^2 + C = -4rg \sin^2(\tau) + C \quad (3.3-8)$$

Das Objekt beginnt seine Bewegung in dem Punkt, in dem  $\tau = \tau_0$  gilt. Die Geschwindigkeit in diesem Punkt soll null sein. Wenn das Objekt im höher gelegenen Punkt A startet, vgl. (3.3-5), haben wir  $\theta = \pi$  und  $\tau = \theta/2 = \pi/2$ . Im tiefsten Punkt ist  $\tau = 0$ . Damit ist die Integrationskonstante C festgelegt, und wir erhalten

$$v^2 = 4rg(\sin^2(\tau_0) - \sin^2(\tau)) \quad (3.3-9)$$

$v$  hängt demnach von der Position der Masse ab. Wir setzen  $\tau_0 = \pi/2$  und berechnen  $T = 4 \int ds/v$ :

$$T = 4 \int_0^{\pi/2} \frac{4r \cos(\tau) d\tau}{\sqrt{4rg \cos(\tau)}} = 2\pi \sqrt{\frac{4r}{g}} \quad (3.3-10)$$

Wir sehen also, dass die Periode des Zykloidenpendels nicht von der Amplitude abhängt.

$T/4 = \pi(r/g)^{1/2}$  ist die Zeit, die eine Masse benötigt, um zum tiefsten Punkt der Zykloide zu gelangen. Unter allen Kurven, die den höchsten mit dem tiefsten Punkt verbinden, existiert keine, die schneller wäre. Für  $r = 1m$  und  $g = 9.81m/s^2$  ergibt sich  $T/4 \approx 1.003s$ .

Wir können diese Zeit vergleichen mit der Zeit, die eine Masse benötigt, die reibungsfrei eine schiefe Ebene mit  $h = 2r = 2m$  und  $\tan(\alpha) = 2/\pi$  hinabgleitet.

Mit

$$t = \frac{\sqrt{2gh}}{g \sin(\alpha)}$$

erhalten wir  $t = 1,189s$ , also eine um  $0.186s$  größere Zeit. (Ein Körper, der reibungsfrei auf einer schiefen Ebene gleitet, unterliegt den Kräften  $\vec{N}$  und  $m\vec{g}$ . Die Komponente der Gewichtskraft parallel zur Ebene ist  $mg \sin(\alpha)$  und erzeugt die Beschleunigung  $a = g \sin(\alpha)$ . Benutzen wir noch  $v = at$  und  $v = (2gh)^{1/2}$ , so erhalten wir die angegebene Formel für die Zeit zum Hinabgleiten.)